

QC — Quality control charts

Description	Quick start	Menu	Syntax
Options	Remarks and examples	Stored results	Methods and formulas
References	Also see		

Description

These commands provide standard quality-control charts. `cchart` draws a *c* chart; `pchart`, a *p* (fraction-defective) chart; `rchart`, an *R* (range or dispersion) chart; `xchart`, an \bar{X} (control line) chart; and `shewhart`, vertically aligned \bar{X} and *R* charts.

Quick start

c chart for `dvar` defects per unit identified by `uvar`

```
cchart dvar uvar
```

p chart for `dvar` defective items out of `nvar` items inspected from each unit identified by `uvar`

```
pchart dvar uvar nvar
```

Same as above, but stabilize the *p* chart for unequal numbers of items inspected per unit

```
pchart dvar uvar nvar, stabilized
```

R chart for the range of measurements `m1`, `m2`, `m3`, and `m4`

```
rchart m1 m2 m3 m4, connect(1)
```

Same as above, but use known process standard deviation of 0.5 for control limits

```
rchart m1 m2 m3 m4, connect(1) std(.5)
```

\bar{X} chart for measurements `m1`, `m2`, `m3`, and `m4`

```
xchart m1 m2 m3 m4, connect(1)
```

Same as above, but use known process standard deviation of 0.5 and grand mean of 10 for control limits

```
xchart m1 m2 m3 m4, connect(1) std(.5) mean(10)
```

Shewhart chart with vertically aligned *R* and \bar{X} charts

```
shewhart m1 m2 m3 m4, connect(1) std(.5) mean(10)
```

Menu

cchart

Statistics > Other > Quality control > C chart

pchart

Statistics > Other > Quality control > P chart

rchart

Statistics > Other > Quality control > R chart

xchart

Statistics > Other > Quality control > X-bar chart

shewhart

Statistics > Other > Quality control > Vertically aligned X-bar and R chart

Syntax

Draw a *c* chart

```
cchart defect_var unit_var [ , cchart_options ]
```

Draw a *p* (fraction-defective) chart

```
pchart reject_var unit_var ssize_var [ , pchart_options ]
```

Draw an *R* (range or dispersion) chart

```
rchart varlist [if] [in] [ , rchart_options ]
```

Draw an \bar{X} (control line) chart

```
xchart varlist [if] [in] [ , xchart_options ]
```

Draw vertically aligned \bar{X} and *R* charts

```
shewhart varlist [if] [in] [ , shewhart_options ]
```

<i>cchart_options</i>	Description
Main	
<code><u>nograph</u></code>	suppress graph
Plot	
<code><i>connect_options</i></code>	affect rendition of the plotted points
<code><i>marker_options</i></code>	change look of markers (color, size, etc.)
<code><i>marker_label_options</i></code>	add marker labels; change look or position
Control limits	
<code><u>clopts</u>(<i>cline_options</i>)</code>	affect rendition of the control limits
Add plots	
<code><u>addplot</u>(<i>plot</i>)</code>	add other plots to the generated graph
Y axis, X axis, Titles, Legend, Overall	
<code><i>twoway_options</i></code>	any options other than <code>by()</code> documented in [G-3] <code><i>twoway_options</i></code>

<i>pchart_options</i>	Description
Main	
<u>stabilized</u>	stabilize the p chart when sample sizes are unequal
<u>nograph</u>	suppress graph
<u>generate</u> (<i>newvar_f</i> <i>newvar_{lcl}</i> <i>newvar_{ucl}</i>)	store the fractions of defective elements and the lower and upper control limits
Plot	
<i>connect_options</i>	affect rendition of the plotted points
<i>marker_options</i>	change look of markers (color, size, etc.)
<i>marker_label_options</i>	add marker labels; change look or position
Control limits	
<u>clopts</u> (<i>cline_options</i>)	affect rendition of the control limits
Add plots	
<u>addplot</u> (<i>plot</i>)	add other plots to the generated graph
Y axis, X axis, Titles, Legend, Overall	
<i>twoway_options</i>	any options other than <code>by()</code> documented in [G-3] <i>twoway_options</i>

<i>rchart_options</i>	Description
Main	
<u>std</u> (#)	user-specified standard deviation
<u>nograph</u>	suppress graph
Plot	
<i>connect_options</i>	affect rendition of the plotted points
<i>marker_options</i>	change look of markers (color, size, etc.)
<i>marker_label_options</i>	add marker labels; change look or position
Control limits	
<u>clopts</u> (<i>cline_options</i>)	affect rendition of the control limits
Add plots	
<u>addplot</u> (<i>plot</i>)	add other plots to the generated graph
Y axis, X axis, Titles, Legend, Overall	
<i>twoway_options</i>	any options other than <code>by()</code> documented in [G-3] <i>twoway_options</i>

<i>xchart_options</i>	Description
Main	
<u>std</u> (#)	user-specified standard deviation
<u>mean</u> (#)	user-specified mean
<u>lower</u> (#) <u>upper</u> (#)	lower and upper limits of the X-bar limits
<u>nograph</u>	suppress graph
Plot	
<i>connect_options</i>	affect rendition of the plotted points
<i>marker_options</i>	change look of markers (color, size, etc.)
<i>marker_label_options</i>	add marker labels; change look or position
Control limits	
<u>clopts</u> (<i>cline_options</i>)	affect rendition of the control limits
Add plots	
<u>addplot</u> (<i>plot</i>)	add other plots to the generated graph
Y axis, X axis, Titles, Legend, Overall	
<i>twoway_options</i>	any options other than by() documented in [G-3] <i>twoway_options</i>

<i>shewhart_options</i>	Description
Main	
<u>std</u> (#)	user-specified standard deviation
<u>mean</u> (#)	user-specified mean
<u>nograph</u>	suppress graph
Plot	
<i>connect_options</i>	affect rendition of the plotted points
<i>marker_options</i>	change look of markers (color, size, etc.)
<i>marker_label_options</i>	add marker labels; change look or position
Control limits	
<u>clopts</u> (<i>cline_options</i>)	affect rendition of the control limits
Y axis, X axis, Titles, Legend, Overall	
<i>combine_options</i>	any options documented in [G-2] graph combine

`collect` is allowed with all QC commands; see [U] **11.1.10 Prefix commands**.

Options

Main

`stabilized` stabilizes the p chart when sample sizes are unequal.

`std(#)` specifies the standard deviation of the process. The R chart is calculated (based on the range) if this option is not specified.

`mean(#)` specifies the grand mean, which is calculated if not specified.

`lower(#)` and `upper(#)` must be specified together or not at all. They specify the lower and upper limits of the \bar{X} chart. Calculations based on the mean and standard deviation (whether specified by option or calculated) are used otherwise.

`nograph` suppresses the graph.

`generate(newvarf newvarlcl newvarucl)` stores the plotted values in the p chart. `newvarf` will contain the fractions of defective elements; `newvarlcl` and `newvarucl` will contain the lower and upper control limits, respectively.

Plot

`connect_options` affect whether lines connect the plotted points and the rendition of those lines; see [G-3] [connect_options](#).

`marker_options` affect the rendition of markers drawn at the plotted points, including their shape, size, color, and outline; see [G-3] [marker_options](#).

`marker_label_options` specify if and how the markers are to be labeled; see [G-3] [marker_label_options](#).

Control limits

`clopts(cline_options)` affects the rendition of the control limits; see [G-3] [cline_options](#).

Add plots

`addplot(plot)` provides a way to add other plots to the generated graph. See [G-3] [addplot_option](#).

Y axis, X axis, Titles, Legend, Overall

`twoway_options` are any of the options documented in [G-3] [twoway_options](#), excluding `by()`. These include options for titling the graph (see [G-3] [title_options](#)) and for saving the graph to disk (see [G-3] [saving_option](#)).

`combine_options` (`shewhart` only) are any of the options documented in [G-2] [graph_combine](#). These include options for titling the graph (see [G-3] [title_options](#)) and for saving the graph to disk (see [G-3] [saving_option](#)).

Remarks and examples

[stata.com](http://www.stata.com)

Control charts may be used to define the goal of a repetitive process, to control that process, and to determine if the goal has been achieved. Walter A. Shewhart of Bell Telephone Laboratories devised the first control chart in 1924. In 1931, Shewhart published *Economic Control of Quality of Manufactured Product*. According to Burr, “Few fields of knowledge have ever been so completely explored and charted in the first exposition” (1976, 29). Shewhart states that “a phenomenon will be said to be controlled when, through the use of past experience, we can predict, at least within limits, how the phenomenon may be expected to vary in the future. Here it is understood that prediction within limits means that we can state, at least approximately, the probability that the observed phenomenon will fall within given limits” (1931, 6).

For more information on quality-control charts, see Burr (1976), Duncan (1986), Harris (1999), or Ryan (2011).

Example 1: cchart

`cchart` graphs a *c* chart showing the number of nonconformities in a unit, where *defect_var* records the number of defects in each inspection unit and *unit_var* records the unit number. The unit numbers need not be in order. For instance, consider the following example dataset from Ryan (2011, 186):

```
. use https://www.stata-press.com/data/r18/ncu
. describe
Contains data from https://www.stata-press.com/data/r18/ncu.dta
Observations:      30
Variables:         2                2 Dec 2022 15:15
```


Variable name	Storage type	Display format	Value label	Variable label
day	byte	%9.0g		Day in April
defects	byte	%9.0g		Number of nonconforming units

Sorted by:

```
. list in 1/5
```

	day	defects
1.	1	7
2.	2	5
3.	3	11
4.	4	13
5.	5	9

```
. cchart defects day, title(c chart for nonconforming transistors)
```



The expected number of defects is 10.6, with lower and upper control limits of 0.8327 and 20.37, respectively. No units are out of control.

► Example 2: pchart

`pchart` graphs a p chart, which shows the fraction of nonconforming items in a subgroup, where `reject_var` records the number rejected in each inspection unit, `unit_var` records the inspection unit number, and `ssize_var` records the number inspected in each unit.

Consider the example dataset from Ryan (2011, 186) of the number of nonconforming transistors out of 1,000 inspected each day during the month of April:

```
. use https://www.stata-press.com/data/r18/ncu2
. describe
```

Contains data from <https://www.stata-press.com/data/r18/ncu2.dta>

```
Observations:      30
Variables:          3                2 Dec 2022 15:16
```


Variable name	Storage type	Display format	Value label	Variable label
day	byte	%9.0g		Day in April
rejects	byte	%9.0g		Number of nonconforming units
ssize	int	%9.0g		Sample size

Sorted by:

```
. list in 1/5
```

	day	rejects	ssize
1.	1	7	1000
2.	2	5	1000
3.	3	11	1000
4.	4	13	1000
5.	5	9	1000

```
. pchart rejects day ssize
```




All the points are within the control limits, which are 0.0009 for the lower limit and 0.0203 for the upper limit.

Here the sample sizes are fixed at 1,000, so the `ssize` variable contains 1,000 for each observation. Sample sizes need not be fixed, however. Say that our data were slightly different:

```
. use https://www.stata-press.com/data/r18/ncu3
. list in 1/5
```


	day	rejects	ssize
1.	1	7	920
2.	2	5	920
3.	3	11	920
4.	4	13	950
5.	5	9	950

```
. pchart rejects day ssize
```



Here the control limits are, like the sample size, no longer constant. The `stabilize` option will stabilize the control chart:

```
. pchart rejects day ssize, stabilize
```



Example 3: rchart


`rchart` displays an R chart showing the range for repeated measurements at various times. Variables within observations record measurements. Observations represent different samples.

For instance, say that we take five samples of 5 observations each. In our first sample, our measurements are 10, 11, 10, 11, and 12. The data are

```
. list
```


	m1	m2	m3	m4	m5
1.	10	11	10	11	12
2.	12	10	9	10	9
3.	10	11	10	12	10
4.	9	9	9	10	11
5.	12	12	12	12	13

```
. rchart m1-m5, connect(1)
```



The expected range in each sample is 2 with lower and upper control limits of 0 and 4.23, respectively. If we know that the process standard deviation is 0.3, we could specify

```
. rchart m1-m5, connect(1) std(.3)
```




◀

► Example 4: xchart


`xchart` graphs an \bar{X} chart for repeated measurements at various times. Variables within observations record measurements, and observations represent different samples. Using the same data as in the [previous example](#), we type

```
. xchart m1-m5, connect(1)
```




The average measurement in the sample is 10.64, and the lower and upper control limits are 9.486 and 11.794, respectively. Suppose that we knew from prior information that the mean of the process is 11. Then, we would type

```
. xchart m1-m5, connect(1) mean(11)
```




If we also know that the standard deviation of the process is 0.3, we could type

```
. xchart m1-m5, connect(1) mean(11) std(.3)
```



Finally, xchart allows us to specify our own control limits:

```
. xchart m1-m5, connect(1) mean(11) lower(10) upper(12)
```




Example 5: shewhart

`shewhart` displays a vertically aligned \bar{X} and R chart in the same image. To produce the best-looking combined image possible, you will want to use the `xchart` and `rchart` commands separately and then combine the graphs. `shewhart`, however, is more convenient.

Using the same data as previously, but realizing that the standard deviation should have been 0.4, we type

```
. shewhart m1-m5, connect(1) mean(11) std(.4)
```



Walter Andrew Shewhart (1891–1967) was born in Illinois and educated as a physicist, with degrees from the Universities of Illinois and California. After a brief period teaching physics, he worked for the Western Electric Company and (from 1925) the Bell Telephone Laboratories. His name is most associated with control charts used in quality controls, but his many other interests ranged generally from quality assurance to the philosophy of science.

Stored results

`cchart` stores the following in `r()`:

Scalars

<code>r(cbar)</code>	expected number of nonconformities
<code>r(lcl_c)</code>	lower control limit
<code>r(uc1_c)</code>	upper control limit
<code>r(N)</code>	number of observations
<code>r(out_c)</code>	number of units out of control
<code>r(below_c)</code>	number of units below the lower limit
<code>r(above_c)</code>	number of units above the upper limit

`pchart` stores the following in `r()`:

Scalars

<code>r(pbar)</code>	average fraction of nonconformities
<code>r(lcl_p)</code>	lower control limit
<code>r(uc1_p)</code>	upper control limit
<code>r(N)</code>	number of observations
<code>r(out_p)</code>	number of units out of control
<code>r(below_p)</code>	number of units below the lower limit
<code>r(above_p)</code>	number of units above the upper limit

`rchart` stores the following in `r()`:

Scalars

<code>r(central_line)</code>	ordinate of the central line
<code>r(lcl_r)</code>	lower control limit
<code>r(uc1_r)</code>	upper control limit
<code>r(N)</code>	number of observations
<code>r(out_r)</code>	number of units out of control
<code>r(below_r)</code>	number of units below the lower limit
<code>r(above_r)</code>	number of units above the upper limit

`xchart` stores the following in `r()`:

Scalars

<code>r(xbar)</code>	grand mean
<code>r(lcl_x)</code>	lower control limit
<code>r(uc1_x)</code>	upper control limit
<code>r(N)</code>	number of observations
<code>r(out_x)</code>	number of units out of control
<code>r(below_x)</code>	number of units below the lower limit
<code>r(above_x)</code>	number of units above the upper limit

`shewhart` stores in `r()` the combination of stored results from `xchart` and `rchart`.

Methods and formulas

For the c chart, the number of defects per unit, C , is taken to be a value of a random variable having a Poisson distribution. If k is the number of units available for estimating λ , the parameter of the Poisson distribution, and if C_i is the number of defects in the i th unit, then λ is estimated by $\bar{C} = \sum_i C_i/k$. Then

$$\begin{aligned}\text{central line} &= \bar{C} \\ \text{UCL} &= \bar{C} + 3\sqrt{\bar{C}} \\ \text{LCL} &= \bar{C} - 3\sqrt{\bar{C}}\end{aligned}$$

Control limits for the p chart are based on the sampling theory for proportions, using the normal approximation to the binomial. If k samples are taken, the estimator of p is given by $\bar{p} = \sum_i \hat{p}_i/k$, where $\hat{p}_i = x_i/n_i$, and x_i is the number of defects in the i th sample of size n_i . The central line and the control limits are given by

$$\begin{aligned}\text{central line} &= \bar{p} \\ \text{UCL} &= \bar{p} + 3\sqrt{\bar{p}(1-\bar{p})/n_i} \\ \text{LCL} &= \bar{p} - 3\sqrt{\bar{p}(1-\bar{p})/n_i}\end{aligned}$$

Control limits for the R chart are based on the distribution of the range of samples of size n from a normal population. If the standard deviation of the process, σ , is known,

$$\begin{aligned}\text{central line} &= d_2\sigma \\ \text{UCL} &= D_2\sigma \\ \text{LCL} &= D_1\sigma\end{aligned}$$

where d_2 , D_1 , and D_2 are functions of the number of observations in the sample and are obtained from the table published in [Beyer \(1976\)](#).

When σ is unknown,

$$\begin{aligned}\text{central line} &= \bar{R} \\ \text{UCL} &= (D_2/d_2)\bar{R} \\ \text{LCL} &= (D_1/d_2)\bar{R}\end{aligned}$$

where $\bar{R} = \sum_i R_i/k$ is the range of the k sample ranges R_i .

Control limits for the \bar{X} chart are given by

$$\begin{aligned}\text{central line} &= \bar{x} \\ \text{UCL} &= \bar{x} + (3/\sqrt{n})\sigma \\ \text{LCL} &= \bar{x} - (3/\sqrt{n})\sigma\end{aligned}$$

if σ is known. If σ is unknown,

$$\text{central line} = \bar{x}$$

$$\text{UCL} = \bar{x} + A_2\bar{R}$$

$$\text{LCL} = \bar{x} - A_2\bar{R}$$

where \bar{R} is the average range as defined above and A_2 is a function (op. cit.) of the number of observations in the sample.

Isobel Loutit (1909–2009) is known for her work during World War II to improve the accuracy of targeting for anti-aircraft guns and as a contributor to the field of quality control. Loutit was born in Selkirk Manitoba, Canada. She graduated from the University of Manitoba in 1929 with a degree in mathematics and was one of the first women to work as a professional statistician in Canada. After graduation, she obtained a job teaching French. However, because of her training, she served as a substitute math teacher when needed.

When World War II started, Loutit took a job as a quality control statistician at Northern Electric. Shortly after that, the Canadian government advertised for women with technical training to fill jobs that had been vacated by men who had gone to war. She took a position testing equipment for the military. This job eventually returned her to Northern Electric, this time as a government employee verifying the accuracy of Northern Electric's Vickers anti-aircraft gun predictor, which was used to aim artillery at incoming planes. Recognizing the quality of her work, the CEO of Northern Electric rehired Loutit as an engineer, the only position for which her pay would not be capped as a woman. She later became the first female manager at Northern Electric and the first woman to chair the Montreal Section of the American Society for Quality Control.

References

- Alejo, J., A. K. Bera, A. F. Galvao, G. Montes-Rojas, and Z. Xiao. 2016. [Tests for normality based on the quantile-mean covariance](#). *Stata Journal* 16: 1039–1057.
- Bayart, D. 2001. Walter Andrew Shewhart. In *Statisticians of the Centuries*, ed. C. C. Heyde and E. Seneta, 398–401. New York: Springer.
- Beyer, W. H. 1976. Factors for computing control limits. In Vol. 2 of *Handbook of Tables for Probability and Statistics*, ed. W. H. Beyer, 451–465. Cleveland, OH: The Chemical Rubber Company.
- Burr, I. W. 1976. *Statistical Quality Control Methods*. New York: Dekker.
- Caulcutt, R. 2004. Control charts in practice. *Significance* 1: 81–84. <https://doi.org/10.1111/j.1740-9713.2004.024.x>.
- Duncan, A. J. 1986. *Quality Control and Industrial Statistics*. 5th ed. Homewood, IL: Irwin.
- Harris, R. L. 1999. *Information Graphics: A Comprehensive Illustrated Reference*. New York: Oxford University Press.
- Ryan, T. P. 2011. *Statistical Methods for Quality Improvement*. 3rd ed. Hoboken, NJ: Wiley.
- Shewhart, W. A. 1931. *Economic Control of Quality of Manufactured Product*. New York: Van Nostrand.

Also see

[R] [serrbar](#) — Graph standard error bar chart

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).